==========**Self_Heating_NPN**=======================

TWEAKING THERMAL BEHAVIORAL MODELS TO MATCH THE ACTUAL MEASURED DATA.



Self_Heating_NPN

```
*
*
*                   /                  H                    /
*              ____/_____________________________________/______
*             /.....................................................,/
*            /.................oxide................................/
*        ___ /....____________________________________________......./
*           /..../   n40x5x1                              /...../
*          /..../   ________________________            /...../
*         /..../ /                          /         /...../
*     W  /..../ / /_____ B /    /_____ E  / /    /_____ C /  /...../
*       /..../ / //___/| /    //___/|  / /    //___/| /  /...../
*      /..../ / / |___|//    / |___|/ / /    / |___|//  /...../
*   __/____/_/_/______/__/_______/_/____/_______/__/______/___
*     |....| |  | p+imp |   |  n+imp  | |    | n+imp |  |.....
*     |....|  \ \______/base\______ _/ /      \______/   |.....    D
*     |....|   \_______________________/                 |.....
*     |....|  _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ |.....
*     |....|/   N BURIED LAYER                          \|.....
*     |.....\___________________________________________/.....  ___
*     |.....................................................|
*     |...............oxide.................................|   T
*     |_____________________________________________________|___
*
*
*     H = 81u   W = 36u   D = 4u  T = 1u
*
*     Therm_Conduct_SiO2     1/Rho     = 1.4 W/(m*K)
*     Spec_heat Silicon      Cv        = 760 J/Kg-K
*     Density Silicon                  = 2.42 gm/cm3
*
*                            Rth_calc = 170degC/W
*                            Cth_calc = 21e-9 J/K
*
*
*                                  BPWR
*    ___________
*    |   VCC    |
*    |  10V     |               I = (v(vcc)-v(ve))*i(ve)
*    |          |
* /VCC\         |            BPWR   VTEMP       RTH
* \___/    B    | C          ____________/\  /\  /\_
*   |       ____|'QN1       _|_         |   \/  \/   |
*  _|_     |    |`-> EM    / _ \       _|_          _|_
*  \\\     |     _|_      \/ \/       ___   CTH     ///
*          |    / _ \     /\_/\         |
*         _|_  | VE |     \___/        _|_
*        /VIN\  \___/       |          \\\
*        \___/   _|_ EM2   _|_
*          |    / _ \      \\\
*         _|_  | BTH |                        BTH
*         \\\   \___/ OUT
*                 |      |OUT|    v = -2.2m*VTEMP
*                 |______|___|
*                 |
*                 | 100           LabMeasured  47mVpk @18uS tau
*                 |____/\  ______  (Rth, Cth adjusted to data)
*                       \/       _|_
*                  0->1.6V       ///
*
VCC     VCC   0      DC 10
VIN     VIN   0      PULSE( 0 2.8 1n 1n 1n 100u 200u )
Q1      VCC   VIN    EM     npnv     1
VE      EM    EM2    DC 0
BTH     EM2   OUT    v = 2.2m*v(VTEMP)
R1      OUT   0      100
BPWR    VTEMP 0      i = (v(VCC) -v(OUT))*i(VE)
RTH     VTEMP 0      160
CTH     VTEMP 0      60n
.TRAN  1u 200u 0 1u
```

```
.control
set    pensize = 2
run
plot   out em2  ylimit 1.55 1.65
plot   -vtemp
.endc
.model      npnv          npn (
*=========================================================
+IS=1.1E-16    NF=1.0        BF=120       VAF=30       IKF=6e-03
+              NR=1.0        BR=0.5       VAR=4        IKR=3e-04
+ISE=9E-17     NE=2
+ISC=1E-21     NC=2
+RB=150        RBM=150                                 IRB=8E-04
+RE=17         RC=110
*=========================================================
+CJE=2E-14     VJE=0.65      MJE=0.3
+CJC=2E-14     VJC=0.65      MJC=0.3
+CJS=3E-14     VJS=0.35      MJS=0.19     XCJC=0.42
+TF=2E-11      XTF=1.25      VTF=1        ITF=0.0035
+TR=6E-09      FC=0.9        PTF=210
*=========================================================
+KF=1.0E-16    AF=1
+XTB=1.4       EG=1.11       XTI=8        TNOM=25
.end

*====================END_OF_SPICE=======================
```

To Covert PDF to plain text click below
http://www.fileformat.info/convert/doc/pdf2txt.htm

The out waveform below was taken in the lab. Silicon Oxide has ten times more thermal insulation compared to silicon



The circuit and output waveforms for the simulation have been adjusted to match real silicon data.





The thermal resistance and change in temperature appear to be very close to the calculated value.

The thermal capacitance appears to be about three times larger than expected for just the silicon's volume. The silicon oxide' thermal capacitance is not being included.



H = 81u W = 36u D = 4u T = 1u

Therm_Conduct_SiO2 1/Rho = 1.4 W/(m*K)
Spec_heat Silicon Cv = 760 J/Kg-K
Density Silicon = 2.42 gm/cm3

Rth_calc = 170degC/W
Cth_calc = 21e-9 J/K

The effects of metal were also not included. Some pretty thick metal was used to hook up this transistor.

A single min geometry metal trace will add this to the thermal circuit.



But since there are three thick and long metal traces connecting the transistor to its pads, the thermal situation is perhaps more like this.



The measured overall Rth was 160degC/W and the measured overall Cth appears to be 60nJ/w. This situation suggests that a little more care may have been needed in laying out the metal traces to this thermal self heating geometry.

```
*#1============WinSpiceVersion================

Self_Heating_NPN

*
*
*                    /                H                        /
*               ____/_____________________________________/______
*              /..................................................../
*             /.................oxide............................./
*        ___ /...._______________________________________....../
*           /..../   n40x5x1                             /...../
*          /..../  ______________________               /...../
*         /..../ / ________    _________ /    ________  /...../
*    W   /..../ / /____ B /   /____  E  / /   /____  C /  /...../
*       /..../ / //___/| /   //___/|  / /   //___/| /  /...../
*      /..../ / / |___|//   / |___|/ / /   / |___|//  /...../
*   __/____/_/_/______/___/________/_/____/______/__/_____/___
*     |....| | | p+imp |   |  n+imp  | |     | n+imp |  |.....|
*     |....|  \ \_____/base\_____ _/ /       \_____/   |.....|
*     |....|   \____________________/                   |.....|   D
*     |....|   _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _   |.....|
*     |....|/   N BURIED LAYER                       \|.....|
*     |.....\_______________________________________/......|___
*     |.....................................................|
*     |................oxide................................|  T
*     |_____________________________________________________|___
*
*
*     H = 81u  W = 36u  D = 4u T = 1u
*
*     Therm_Conduct_SiO2    1/Rho    = 1.4 W/(m*K)
*     Spec_heat Silicon     Cv       = 760 J/Kg-K
*     Density Silicon                = 2.42 gm/cm3
*
*                           Rth_calc = 170degC/W
*                           Cth_calc = 21e-9 J/K
*
*
*    ___________                        BPWR
*   |    VCC    |
*   |   10V     |                    I = (v(vcc)-v(ve))*i(ve)
*   |           |
* /VCC\         |               BPWR    VTEMP      RTH
* \___/   B    _| C               _________/\  /\  /\_
*   |      ____|'QN1              _|_      |    \/  \/   |
*  _|_    |    |`-> EM           / _ \    _|_           _|_
*  \\\    |    _|_               \/ \/    ___   CTH      ///
*         |   /   \              /\_/\     |
*        _|_ | VE  |             \___/    _|_
*       /VIN\ \___/                |      \\\
*       \___/  _|_ EM2            _|_
*         |   /   \               \\\
*        _|_ | BTH |                            BTH
*        \\\  \___/ OUT         ___
*               |_______|OUT|   v = -2.2m*VTEMP
*               |       |___|
*               | 100           LabMeasured  47mVpk @18uS tau
*               |____/\  ______  (Rth, Cth adjusted to data)
*                      \/      _|_
*                0->1.6V       ///
*
VCC     VCC   0      DC 10
VIN     VIN   0      PULSE( 0 2.8 1n 1n 1n 100u 200u )
Q1      VCC   VIN    EM     npnv    1
VE      EM    EM2    DC 0
BTH     EM2   OUT    v = 2.2m*v(VTEMP)
R1      OUT   0      100
BPWR    VTEMP 0      i = (v(VCC) -v(OUT))*i(VE)
RTH     VTEMP 0      160
CTH     VTEMP 0      60n
.TRAN  1u 200u 0 1u

.control
set    pensize = 2
run
plot   out em2  ylimit 1.55 1.65
plot   -vtemp
.endc
.model     npnv         npn (
*===========================================================
+IS=1.1E-16   NF=1.0       BF=120      VAF=30      IKF=6e-03
+             NR=1.0       BR=0.5      VAR=4       IKR=3e-04
+ISE=9E-17    NE=2
+ISC=1E-21    NC=2
+RB=150       RBM=150                              IRB=8E-04
+RE=17        RC=110
*===========================================================
+CJE=2E-14    VJE=0.65     MJE=0.3
+CJC=2E-14    VJC=0.65     MJC=0.3
+CJS=3E-14    VJS=0.35     MJS=0.19    XCJC=0.42
+TF=2E-11     XTF=1.25     VTF=1       ITF=0.0035
+TR=6E-09     FC=0.9       PTF=210
*===========================================================
+KF=1.0E-16   AF=1
+XTB=1.4      EG=1.11      XTI=8      TNOM=25
.end
```